Math 22 Fall 2004

Linear Algebra with Applications

Crash Course in Doing Linear Algebra with Maple

September 29, 2004

Load the package for doing Linear Algebra

```
> with(LinearAlgebra):
```

Column vectors and row vectors

```
> v1 := <1|2|3|4>:
  v2 := <5,4,3,1>:
  v3 := Vector[row](5):
  v4 := Vector(4, symbol = x):
  v5 := Vector[row](5, i -> sin(Pi / i)):
  v1, v2, v3, v4, v5;
```

$$[1, 2, 3, 4], \begin{bmatrix} 5 \\ 4 \\ 3 \\ 1 \end{bmatrix}, [0, 0, 0, 0, 0], \begin{bmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \end{bmatrix}, \left[0, 1, \frac{1}{2}\sqrt{3}, \frac{1}{2}\sqrt{2}, \sin\left(\frac{1}{5}\pi\right) \right]$$

Matrix is a row of column vectors or a column of row vectors

```
> M1 := <<1,2,3,4> | <3,4,5,6> | <5,6,7,8>>:
  M2 := <<3|2|1>, <-4|3|2>, <5|3|-1>, <6|-2|4>>:
  u1 := Vector(3, symbol = x): u2 := Vector(3, symbol = y):
  u3 := Vector(3, symbol = z): u4 := Vector([4, -2, 1]):
  M3 := <u1 | u2 | u3 | u4>:
  M4 := Matrix(4, 3, symbol = a):
  M1, M2, M3, M4;
```

$$\begin{bmatrix} 1 & 3 & 5 \\ 2 & 4 & 6 \\ 3 & 5 & 7 \\ 4 & 6 & 8 \end{bmatrix}, \begin{bmatrix} 3 & 2 & 1 \\ -4 & 3 & 2 \\ 5 & 3 & -1 \\ 6 & -2 & 4 \end{bmatrix}, \begin{bmatrix} x_1 & y_1 & z_1 & 4 \\ x_2 & y_2 & z_2 & -2 \\ x_3 & y_3 & z_3 & 1 \end{bmatrix}, \begin{bmatrix} a_{1,1} & a_{1,2} & a_{1,3} \\ a_{2,1} & a_{2,2} & a_{2,3} \\ a_{3,1} & a_{3,2} & a_{3,3} \\ a_{4,1} & a_{4,2} & a_{4,3} \end{bmatrix}$$

Elements of vectors and matrices can be accessed by their indices

```
> v5[2]; M3[2, 3]; M4[3, 2]; v5[5] := evalf(v5[5]): v5;
```

$$1$$

$$z_2$$

$$a_{3,2}$$

$$\left[0, 1, \frac{1}{2}\sqrt{3}, \frac{1}{2}\sqrt{2}, 0.5877852524\right]$$

We can apply a map to the whole vector or matrix

```
> Map(exp, v1), Map(x -> x^2, M3);
```

$$\left[\mathbf{e}, \mathbf{e}^2, \mathbf{e}^3, \mathbf{e}^4\right], \begin{bmatrix} x_1^2 & y_1^2 & z_1^2 & 16 \\ x_2^2 & y_2^2 & z_2^2 & 4 \\ x_3^2 & y_3^2 & z_3^2 & 1 \end{bmatrix}$$

Operations with vectors and matrices

```
> M1 + 2 * M4, v2.v4, M3.M2;
```

$$\begin{bmatrix} 1 + 2\,a_{1,1} & 3 + 2\,a_{1,2} & 5 + 2\,a_{1,3} \\ 2 + 2\,a_{2,1} & 4 + 2\,a_{2,2} & 6 + 2\,a_{2,3} \\ 3 + 2\,a_{3,1} & 5 + 2\,a_{3,2} & 7 + 2\,a_{3,3} \\ 4 + 2\,a_{4,1} & 6 + 2\,a_{4,2} & 8 + 2\,a_{4,3} \end{bmatrix}, 5\,x_1 + 4\,x_2 + 3\,x_3 + x_4,$$

$$\begin{bmatrix} 3x_1^2 - 4y_1^2 + 5z_1^2 + 96 & 2x_1^2 + 3y_1^2 + 3z_1^2 - 32 & x_1^2 + 2y_1^2 - z_1^2 + 64 \\ 3x_2^2 - 4y_2^2 + 5z_2^2 + 24 & 2x_2^2 + 3y_2^2 + 3z_2^2 - 8 & x_2^2 + 2y_2^2 - z_2^2 + 16 \\ 3x_3^2 - 4y_3^2 + 5z_3^2 + 6 & 2x_3^2 + 3y_3^2 + 3z_3^2 - 2 & x_3^2 + 2y_3^2 - z_3^2 + 4 \end{bmatrix}$$

Conversion from a linear system to a matrix, and back

```
> eqsys := [2*x[1] - x[3] = 5, -x[1] + x[2] + x[3] = -2, 2*x[2]+x[3]=3];
  vars := [x[1], x[2], x[3]];
```

$$eqsys := [2x_1 - x_3 = 5, -x_1 + x_2 + x_3 = -2, 2x_2 + x_3 = 3]$$

$$vars := [x_1, x_2, x_3]$$

```
> A, b := GenerateMatrix(eqsys, vars):
  A1 := GenerateMatrix(eqsys, vars, augmented=true ):
  A, b, A1, <A|b> - A1;
```

$$\begin{bmatrix} 2 & 0 & -1 \\ -1 & 1 & 1 \\ 0 & 2 & 1 \end{bmatrix}, \begin{bmatrix} 5 \\ -2 \\ 3 \end{bmatrix}, \begin{bmatrix} 2 & 0 & -1 & 5 \\ -1 & 1 & 1 & -2 \\ 0 & 2 & 1 & 3 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

```
> M5 := Transpose(M1);
  GenerateEquations(M5, [x, y, z]); GenerateEquations(M5, [x, y, z, w], <10,
  -3, 2>);
```

$$M5 := \begin{bmatrix} 1 & 2 & 3 & 4 \\ 3 & 4 & 5 & 6 \\ 5 & 6 & 7 & 8 \end{bmatrix}$$

$$[x + 2y + 3z = 4, 3x + 4y + 5z = 6, 5x + 6y + 7z = 8]$$

$$[x + 2y + 3z + 4w = 10, 3x + 4y + 5z + 6w = -3, 5x + 6y + 7z + 8w = 2]$$

Transform matrices to echelon and reduced echelon forms

```
> GaussianElimination(A1), GaussianElimination(A1, method=FractionFree),
  ReducedRowEchelonForm(A1);
```

$$\begin{bmatrix} 2 & 0 & -1 & 5 \\ 0 & 1 & \frac{1}{2} & \frac{1}{2} \\ 0 & 0 & 0 & 2 \end{bmatrix}, \begin{bmatrix} 2 & 0 & -1 & 5 \\ 0 & 2 & 1 & 1 \\ 0 & 0 & 0 & 4 \end{bmatrix}, \begin{bmatrix} 1 & 0 & \frac{-1}{2} & 0 \\ 0 & 1 & \frac{1}{2} & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> A[1, 2] := h:
  GaussianElimination(<A | b>);
```

$$\begin{bmatrix} 2 & h & -1 & 5 \\ 0 & 1+\frac{1}{2}h & \frac{1}{2} & \frac{1}{2} \\ 0 & 0 & \frac{h}{2+h} & \frac{4+3h}{2+h} \end{bmatrix}$$

Solve a linear system given by a matrix

```
> LinearSolve(A, b);
  LinearSolve(A1);
```

$$\begin{bmatrix} \frac{5h+2}{h} \\ -\frac{2}{h} \\ \frac{4+3h}{h} \end{bmatrix}$$

```
Error, (in LinearSolve) inconsistent system
```

```
> A1[1, 2] := h:
  LinearSolve(A, b), LinearSolve(A1);
```

$$\left[\begin{bmatrix} \frac{5h+2}{h} \\ -\frac{2}{h} \\ \frac{4+3h}{h} \end{bmatrix}, \begin{bmatrix} \frac{5h+2}{h} \\ -\frac{2}{h} \\ \frac{4+3h}{h} \end{bmatrix}\right]$$

Use help for more help

```
> ?LinearAlgebra
>
```